

Functional Programming

WS 2009/10

Christian Sternagel (VO)
 Friedrich Neurauder (PS) Sarah Winkler (PS)

Computational Logic
 Institute of Computer Science
 University of Innsbruck

16 October 2009



Recursion, Pattern Matching, and Currying

Example

```
let rec map (f,ls) = match ls with
| Empty      -> Empty
| List(x,xs) -> List(f x,map(f,xs))
```

User-defined Types

user-defined (data) constructor with argument

user-defined type constructor

recursive definition

```
type 'a mylist = Empty | List of ('a * 'a mylist)
```

type variable

type variable

type variable

user-defined (data) constructor without argument

Recursion, Pattern Matching, and Currying (cont'd)

Currying

```
let rec map f ls = match ls with
| Empty      -> Empty
| List(x,xs) -> List(f x,map f xs)
```

Syntactic Sugar

```
let rec map f = function
| Empty      -> Empty
| List (x, xs) -> List (f x, map f xs)
```

This Week

Practice I

OCaml introduction, **lists**, strings, trees

Theory I

lambda-calculus, evaluation strategies, induction, reasoning about functional programs

Practice II

efficiency, tail-recursion, combinator-parsing

Theory II

type checking, type inference

Advanced Topics

lazy evaluation, infinite data structures, monads, ...

The Type of Lists

Polymorphic Lists

```
type 'a list = [] | (::) of ('a * 'a list)
```

predefined type infix 'cons'

Example

<code>[true;false]</code>	bool list	<code>[(3,2)]</code>	<code>(int * int)list</code>
<code>[1;3;5;7]</code>	int list	<code>[]</code>	<code>'a list</code>
<code>['a';'b']</code>	char list	<code>2::3::[]</code>	int list
<code>["str";"w"]</code>	string list	<code>'a'::['b']</code>	char list

Accessing List Elements - Selectors

```
let hd = function []    -> failwith "empty_list"
              | x::_ -> x

let tl = function []    -> failwith "empty_list"
              | _::xs -> xs
```

A Polymorphic List Function

Example (Init)

```
let rec init i l = if l < 1 then [] else i::init i (l-1)
```

- ▶ this function has type `'a -> int -> 'a list`
- ▶ hence it is polymorphic in `i`

```
init 'c' 2
→ if 2 < 1 then [] else 'c'::init 'c' (2-1)
→+ 'c'::init 'c' 1
→ 'c'::if 1 < 1 then [] else 'c'::init 'c' (1-1)
→+ 'c'::'c'::init 'c' 0
→ 'c'::'c'::if 0 < 1 then []
                        else 'c'::init 'c' (0-1)
→+ ['c';'c']
```

Functions on Integer Lists

Example (Range, Sum, Prod)

```
let rec range m n = if m > n then [] else m::range (m+1) n

let rec sum = function []    -> 0
                  | x::xs -> x + sum xs

let rec prod = function []    -> 1
                  | x::xs -> x * prod xs
```

Functions on Integer Lists (cont'd)

```
range 1 3 = [1;2;3]
range 3 2 = []
```

```
sum [1;2;3] = 1 + 2 + 3
sum [] = 0
```

```
prod [1;2;3] = 1 * 2 * 3
prod [] = 1
```

$$\text{sum}(\text{range } 1 \ n) = \sum_{i=1}^n i$$

Higher-Order Functions

- functions taking functions as arguments

Example (Map)

```
let rec map f = function []    -> []
                  | x::xs -> f x::map f xs

map succ [1;2;3]
→ succ 1::map succ [2;3]
→ 2::map succ [2;3]
→ 2::succ 2::map succ [3]
→ 2::3::map succ [3]
→ 2::3::succ 3::map succ []
→ 2::3::4::map succ []
→ 2::3::4::[] = [2;3;4]
```

Fold - A Very Expressive Function

```
let rec foldr f xs b = match xs with
  | []    -> b
  | x::xs -> f x (foldr f xs b)
```

```
sum ls = foldr (+) ls 0
prod ls = foldr ( * ) ls 1
```

```
foldr f [e1;e2;...;eN] b = f e1 (f e2 (...(f eN b)...))
```

Structuring Code

Modules are used to ...

- ▶ split source code into several files
- ▶ separate namespaces for functions and types
- ▶ abstract from concrete representations

Module Basics - Split Source Code

- ▶ for each module *Module* create **implementation** file *module.ml*
- ▶ code of each module goes into corresponding *.ml file

Example

```
let hd = ...
let tl = ...
let rec init i l = ...
let rec map f = ...
let rec foldr f b = ...
```

lst.ml

```
let rec range m n = ...
let sum = ...
let prod = ...
```

intLst.ml

Module Basics - Separate Namespaces

- ▶ refer to function *fun* from module *Module* by *Module.fun*
- ▶ no problem to have same function names in different modules

Example

Compute the greatest number that can be encoded in binary using *n* bits.

```
let pow2 n = IntLst.prod(Lst.init 2 n)
let maxbin n = IntLst.sum(
  Lst.map pow2 (IntLst.range 0 (n-1))
);;
Printf.printf "%i\n" (maxbin(read_int()))
```

maxbin.ml

Module Basics - Abstraction

- ▶ create **interface** file *module.mli* for module *Module*
- ▶ **signature** (i.e., names and types) of module goes into corresponding *.mli file

Example

```
type 'a t
val empty : 'a t
val push : 'a -> 'a t -> 'a t
val pop : 'a t -> ('a * 'a t)
```

stck.mli

```
open Lst
type 'a t = 'a list
let empty = []
let push e s = e::s
let pop s = (hd s,tl s)
```

stck.ml

```
type 'a t =
| Empty
| Full of ('a * 'a t)
...
```

stck.ml